
                     TITLE 'DES/370 v1.0,  (c)&(p) Bernd R. Fix 1988-1995'
*
********************************************************************
*                                                                  *
*       PROGRAM-ID.  DES/370 EN-/DECRYPTION FACILITY.              *
*           AUTHOR.  BERND R. FIX.                                 *
*     DATE WRITTEN.  09/16/88.                                     *
*        COPYRIGHT.  -  DES/370 ENCRYPTION FACILITY                *
*                       CONTAINS RESTRICTED MATERIAL OF THE AUTHOR.*
*                       (C) COPYRIGHT BERND R. FIX 1988-1995       *
*                       LICENSED MATERIALS - PROPERTY OF B.R. FIX  *
*                       REFER TO COPYRIGHT INSTRUCTIONS            *
*          REMARKS.  THIS PROGRAM UTILIZES THE DATA ENCRYPTION     *
*                    STANDARD (DES) TO EN-/DECRYPT DATA SETS       *
*                    SPECIFIED WITH A KEY SUPPORTED BY THE USER.   *
*                                                                  *
********************************************************************
*
CIPHER       CSECT
             LR      R12,R15
             USING   CIPHER,R12
             ST      R14,RETURN
             LA      R13,SAVE

             LA      R2,CRLF
             SVC     209
             LA      R2,HELLO
             SVC     209
             LA      R2,CRLF
             SVC     209
             L       R15,=V(PET)
             BALR    R14,R15
             MVI     MODE,0
             LA      R1,LINE1
             LA      R2,MYKEY
             BAL     R14,CONVERT
             LA      R2,CRLF
             SVC     209
             LA      R2,TXTKEY0
             SVC     209
             LA      R2,TXTKEY
             SVC     209
             LA      R11,MYKEY
             BAL     R14,KEYSCHED
             LA      R1,LINE2
             LA      R2,BLOCK
             BAL     R14,CONVERT

             LA      R2,CRLF
             SVC     209
             LA      R2,TXTBLOCK
             SVC     209
             LA      R2,TXTDES1
             SVC     209
             LA      R11,BLOCK
             BAL     R14,DES
             MVC     TEST(8),BLOCK
             MVI     MODE,1
             LA      R1,LINE2
             LA      R2,BLOCK
             BAL     R14,CONVERT
             LA      R2,TXTBLOCK
             SVC     209
             LA      R2,TXTDES2
             SVC     209
             LA      R11,TEST
             BAL     R14,DES
             LA      R1,LINE2
             LA      R2,TEST
             BAL     R14,CONVERT
             LA      R2,TXTBLOCK

             SVC     209
             LA      R2,CRLF
             SVC     209
             L       R15,=V(PET)
             BALR    R14,R15
             LA      R2,CRLF
             SVC     209
             LA      R2,BYE
             SVC     209
             LA      R2,CRLF
             SVC     209
             L       R14,RETURN
             BR      R14                     Programm beenden.
*
SAVE         DS      18F                     Sicherungsbereich.
RETURN       DS      1F
MYKEY        DC      X'8000000000000000'
BLOCK        DC      X'0000000000000000'
TEST         DC      X'0000000000000000'
LINE         DC      16C' ',C'$'
HELLO        DC      C'DES/370  VERSION 1.00  (C) BERND FIX, 1987, 88$'
TXTKEY0      DC      C'DEFINED KEY: &'
LINE1        DC      16X'F0',C'$'

TXTKEY       DC      C'CALL KEYSCHEDULE-ROUTINE...$'
TXTDES1      DC      C'ENCRYPTION OF BLOCK...$'
TXTDES2      DC      C'DECRYPTION OF BLOCK...$'
BYE          DC      C'DES/370 TERMINATED.$'
TXTBLOCK     DC      C'BLOCK: &'
LINE2        DC      16X'F0'
CRLF         DC      C'$'
*
CONVERT      UNPK    0(15,R1),0(8,R2)
             MVC     15(1,R1),14(R1)
             IC      R2,15(R1)
             SRL     R2,4
             STC     R2,15(R1)
             OC      14(2,R1),=X'F0F0'
             TR      0(16,R1),CVTTAB
             BR      R14
CVTTAB       DC      240C'*'
             DC      X'F0F1F2F3F4F5F6F7F8F9C1C2C3C4C5C6'
*
             LTORG
DES          STM     R14,R12,12(R13)         Sichern der Register.
             LA      R2,STACK                Stackbereiche miteinander
             ST      R2,8(R13)               verketten.



       DES/370

        Bernd Fix                         1987                                           6    17

             ST      R13,4(R2)
             LR      R13,R2
             MVC     INBLK(8),0(R11)         Zu kodierenden Block holen.
             BAL     R14,INIPER              Eingangspermutation ausfuehren.
             LA      R8,KS                   R8 = momentaner Schluessel.
             LA      R9,8                    R9 = Offset naechster Schluessel
             TM      MODE,1                  Umsetzungsmodus = Decipher ?
             BZ      $32                     Nein: -->
             LA      R8,120(R8)              Reihenfolge K16 ... K1
             LNR     R9,R9                   => negativer Offset
$32          LA      R10,16                  16 mal die Ver.-Funktion aufr.
             L       R1,OUTBLK               Die Variablen R und L festlegen.
             ST      R1,L
             L       R1,OUTBLK+4
             ST      R1,R
$14          L       R0,R                    R0 = f(R0,Kn) ist die
             BAL     R14,FUNCTION            Verschluesselungsfunktion.
             X       R0,L                    Erg. R0 ex-or mit L
             MVC     L(4),R                  neues L = altes R
             ST      R0,R                    neues R = altes L + f(R,Kn)
             AR      R8,R9                   Naechsten Schluessel.
             BCT     R10,$14                 Schon 16 mal durch? Nein:-->
             L       R1,R                    Den Block fuer die Ausgangs-

             ST      R1,INBLK                permutation nach INBLK bringen.
             L       R1,L
             ST      R1,INBLK+4
             BAL     R14,FINPER
             MVC     0(8,R11),OUTBLK
             L       R13,STACK+4
             LM      R14,R12,12(R13)         Register zurueckholen und
             BR      R14                     Ruecksprung.
STACK        DS      18F                     Rettungsbereich f. Register.
R            DC      F'0'                    Variablen f. Kodierung.
L            DC      F'0'
MODE         DC      X'00'
*
INIPER       LA      R2,8                    Acht Bytes (64 Bits) werden
             XR      R1,R1                   permutiert.
$1           IC      R1,PERTAB-1(R2)         Byte#(PERTAB) wird generiert.
             LA      R3,8                    Acht Bits sind zu bearbeiten.
$2           LA      R6,INBLK-1(R3)          Permutation wird generiert,
             ICM     R5,B'1000',0(R6)        indem in ein Ziel-Byte das jew.
             SLDL    R4,1                    linkeste Bit der 8 Quell-Bytes
             STCM    R5,B'1000',0(R6)        geschoben wird.
             BCT     R3,$2                   Schon alle Bits durch? Nein:-->
             STC     R4,OUTBLK-1(R1)         Speichere Resultat.

             BCT     R2,$1                   Alle Bytes generiert? Nein:-->
             BR      R14                     Fertig.
*
FINPER       LA      R2,8                    Acht Bytes zu bearbeiten in
             XR      R1,R1                   der Schluss-Permutation.
$3           IC      R1,PERTAB-1(R2)         Byte#(PERTAB) wird bearbeitet.
             LA      R6,INBLK-1(R1)
             ICM     R5,B'1000',0(R6)
             LA      R3,8                    Acht Bits pro Durchgang.
$4           IC      R4,OUTBLK-1(R3)         Die 8 Bits des Q-Bytes werden
             SLDL    R4,1                    von links her bitweise in die
             STC     R4,OUTBLK-1(R3)         8 Z-Bytes geschoben.
             BCT     R3,$4                   Schon alle Bits durch? Nein:-->
             BCT     R2,$3                   Schon alle Q-Bytes? Nein:-->
             BR      R14                     Fertig.
*
             DS      0F
INBLK        DC      8X'00'                  Quell-Bytes
OUTBLK       DC      8X'00'                  Ziel-Bytes
PERTAB       DC      X'4,8,3,7,2,6,1,5'      Tabelle fuer FP
*
KEYSCHED     STM     R14,R12,12(R13)         Sichern aller Register.
             MVC     KEY(8),0(R11)

             XR      R1,R1
             LA      R2,7                    Sieben Bytes ist TEMP lang.
$5           IC      R1,TAB3-1(R2)           Davon Byte#(TAB3) berechnen.
             IC      R4,TEMP-1(R1)           TEMP ist das Resultat.
             LA      R3,8                    Acht Bytes ist der KEY lang.
$6           LA      R6,KEY-1(R3)            Die Permuted Choice 1 (PC-1)
             ICM     R5,B'1000',0(R6)        laesst sich auch durch einfaches
             SLDL    R4,1                    bitweises Schieben des jeweils
             STCM    R5,B'1000',0(R6)        linkesten KEY-Byte-Bits nach
             BCT     R3,$6                   TEMP erzeugen. 8 x durchlaufen.
             STC     R4,TEMP-1(R1)           Speichere Resultat-Byte.
             BCT     R2,$5                   Alle Bytes von TEMP erzeugen.
             L       R1,TEMP                 Aus TEMP wird die Variable
             SRL     R1,4                    C berechnet.
             ST      R1,C0                   Speichere die Variable C.
             XR      R1,R1                   Variable D gemaess der Defi-
             XR      R0,R0                   nition berechnen.
             ICM     R1,B'1110',TEMP+4
             SRL     R1,4
             NI      TEMP+3,X'0F'
             IC      R0,TEMP+3
             OR      R1,R0
             ST      R1,D0                   Variable D speichern.

             LA      R2,16                   Insgesamt sind die Schluessel
             LA      R3,KS                   K1 bis K16 (in KS) zu erzeugen.
$7           IC      R4,TAB4-1(R2)           Dazu werden C0 und D0 um (R4)
             XR      R0,R0                   Berechne das rotierte C0.
             L       R1,C0
             SLDL    R0,4(R4)
             SRL     R1,4
             OR      R1,R0
             ST      R1,C0
             XR      R0,R0                   Berechne das rotierte D0.
             L       R1,D0
             SLDL    R0,4(R4)
             SRL     R1,4
             OR      R1,R0
             ST      R1,D0
             XR      R1,R1
             LA      R7,4                    Fuehre die Permuted Choice 2
             LA      R6,24                   aus durch direktes Abfragen
$33          LA      R8,6                    von Bitstellen der Bitstellen-
$34          IC      R1,PC2C-1(R6)           kombination CD.
             L       R0,MASKE
             SRL     R0,3(R1)
             L       R5,MASKE

             IC      R4,WORK-1(R7)
             N       R0,C0
             BNZ     $35
             LA      R5,0
$35          SLDL    R4,1
             STC     R4,WORK-1(R7)
             IC      R1,PC2D-1(R6)
             L       R0,MASKE
             SRL     R0,3(R1)
             L       R5,MASKE
             IC      R4,WORK+3(R7)
             N       R0,D0
             BNZ     $36
             LA      R5,0
$36          SLDL    R4,1
             STC     R4,WORK+3(R7)
             BCTR    R6,0
             BCT     R8,$34
             BCT     R7,$33
             NC      WORK(8),SIGBIT
             MVC     0(8,R3),WORK
             LA      R3,8(R3)                Adressiere naechsten Schluessel.
             BCT     R2,$7                   Alle 16 Schluessel erzeugen.

             LM      R14,R12,12(R13)         Hole Register.
             BR      R14                     Fertig.
*
TAB3         DC      X'5,6,7,4,3,2,1'        Tabelle fuer KeySchedule
TAB4         DC      X'1,2,2,2,2,2,2,1,2,2,2,2,2,2,1,1'
PC2C         DC      AL1(5,1,24,11,17,14,10,21,6,15,28,3)
             DC      AL1(8,26,4,12,19,23,2,13,20,27,7,16)
PC2D         DC      AL1(27,19,9,3,24,13,20,5,17,23,12,2)
             DC      AL1(25,6,28,11,21,16,4,1,8,22,14,18)
MASKE        DC      B'10000000',AL3(0)     Maske fuer Bitstellenabfrage
SIGBIT       DC      8B'00111111'           Signifikante Bits
TEMP         DC      2F'0'                  Resultat-Speicher.
WORK         DC      2F'0'                  Arbeitsspeicher.
C0           DC      F'0'                   Variable
D0           DC      F'0'                   Variable
KEY          DC      2F'0'                  Schluessel fuer DES!
KS           DC      32F'0'                 16x 64-Bit-Schluessel Kn
*
FUNCTION     STM     R14,R12,12(R13)
             LA      R2,8                    Acht Bits pro WORK-Byte.
             LR      R5,R0                   Auf 48 Bit expandieren.
$8           LA      R3,4                    4 Byte ist R regulaer.
$9           IC      R4,WORK(R3)             Schiebe das jeweils linkeste

             SLDL    R4,1                    Bit von R in das naechste
             STC     R4,WORK(R3)             zykl. Byte von WORK.
             BCT     R3,$9                   Alle 4 Bytes mit
             BCT     R2,$8                   jeweils 8 Bit durch? Nein:-->
             XR      R1,R1
             IC      R1,WORK+1               Die beiden "Rand"Bytes von
             LR      R0,R1                   WORK sind um eine Stelle
             SLL     R0,7                    nach links bzw. rechts ge-
             SRL     R1,1                    schobene Bytes des inneren
             OR      R1,R0                   WORK-Feldes.
             STC     R1,WORK+5               Auf diese Weise werden aus
             XR      R1,R1                   den 32 "inneren" Bits durch
             IC      R1,WORK+4               Hinzufuegen von 2 Bytes (16 Bit)
             SLL     R1,1                    die benoetigten 48 Bit.
             LR      R0,R1
             SRL     R0,8
             OR      R1,R0
             STC     R1,WORK
             LA      R2,8                    Die Bits des Feldes WORK
             LA      R1,0                    muessen noch in die nieder-
*                                            wertigsten 6 Bits der Bytes
$10          LA      R3,6                    im Feld TEMP gebracht werden,
$11          LA      R6,WORK-1(R3)           was durch shiften des jeweils

             ICM     R5,B'1000',0(R6)        linkesten Bits eines der 6 Bytes
             SLDL    R4,1                    in ein Byte von TEMP geschieht.
             STCM    R5,B'1000',0(R6)
             BCT     R3,$11
             N       R4,SIGBIT
             STC     R4,TEMP(R1)
             LA      R1,1(R1)
             BCT     R2,$10
             XC      TEMP(8),0(R8)           Exklusiv-Oder Verknuepfung
*                                            zwischen TEMP und Kn !
             XR      R4,R4
             LA      R2,8                    Aus den 48 Bits, die in 8 Bytes
             LA      R3,STAB                 untergebracht sind, werden jetzt
$12          IC      R4,TEMP-1(R2)           nach Tabellen S1 bis S8 jeweils
             IC      R4,0(R4,R3)             6 Bit auf 4 Bit komprimiert.
             SRDL    R4,4                    Das Ergebnis hat dann 4 Bit x 8,
             LA      R3,64(R3)               also genau die benoetigten
             BCT     R2,$12                  32 Bit, die auch R und L lang
             ST      R5,TEMP                 sind.
             LA      R2,32                   Diese 32 Bit werden vor Rueck-
$37          IC      R1,PTAB-1(R2)           kehr zum rufenden Programm noch
             L       R0,MASKE                permutiert.
             SRL     R0,0(R1)

             L       R5,MASKE
             N       R0,TEMP
             BNZ     $38
             LA      R5,0
$38          SLDL    R4,1
             BCT     R2,$37
             ST      R4,TEMP
             LM      R14,R12,12(R13)
             L       R0,TEMP                 In R0 steht das Ergebnis.
             BR      R14                     Fertig.
*
PTAB         DC      AL1(24,3,10,21,5,29,12,18,8,2,26,31,13,23,7,1)
             DC      AL1(9,30,17,4,25,22,14,0,16,27,11,28,20,19,6,15)
STAB         DC      AL1(13,1,2,15,8,13,4,8,6,10,15,3,11,7,1,4)
             DC      AL1(10,12,9,5,3,6,14,11,5,0,0,14,12,9,7,2)
             DC      AL1(7,2,11,1,4,14,1,7,9,4,12,10,14,8,2,13)
             DC      AL1(0,15,6,12,10,9,13,0,15,3,3,5,5,6,8,11)
             DC      AL1(4,13,11,0,2,11,14,7,15,4,0,9,8,1,13,10)
             DC      AL1(3,14,12,3,9,5,7,12,5,2,10,15,6,8,1,6)
             DC      AL1(1,6,4,11,11,13,13,8,12,1,3,4,7,10,14,7)
             DC      AL1(10,9,15,5,6,0,8,15,0,14,5,2,9,3,2,12)
             DC      AL1(12,10,1,15,10,4,15,2,9,7,2,12,6,9,8,5)
             DC      AL1(0,6,13,1,3,13,4,14,14,0,7,11,5,3,11,8)

             DC      AL1(9,4,14,3,15,2,5,12,2,9,8,5,12,15,3,10)
             DC      AL1(7,11,0,14,4,1,10,7,1,6,13,0,11,8,6,13)
             DC      AL1(2,14,12,11,4,2,1,12,7,4,10,7,11,13,6,1)
             DC      AL1(8,5,5,0,3,15,15,10,13,3,0,9,14,8,9,6)
             DC      AL1(4,11,2,8,1,12,11,7,10,1,13,14,7,2,8,13)
             DC      AL1(15,6,9,15,12,0,5,9,6,10,3,4,0,5,14,3)
             DC      AL1(7,13,13,8,14,11,3,5,0,6,6,15,9,0,10,3)
             DC      AL1(1,4,2,7,8,2,5,12,11,1,12,10,4,14,15,9)
             DC      AL1(10,3,6,15,9,0,0,6,12,10,11,1,7,13,13,8)
             DC      AL1(15,9,1,4,3,5,14,11,5,12,2,7,8,2,4,14)
             DC      AL1(10,13,0,7,9,0,14,9,6,3,3,4,15,6,5,10)
             DC      AL1(1,2,13,8,12,5,7,14,11,12,4,11,2,15,8,1)
             DC      AL1(13,1,6,10,4,13,9,0,8,6,15,9,3,8,0,7)
             DC      AL1(11,4,1,15,2,14,12,3,5,11,10,5,14,2,7,12)
             DC      AL1(15,3,1,13,8,4,14,7,6,15,11,2,3,8,4,14)
             DC      AL1(9,12,7,0,2,1,13,10,12,6,0,9,5,11,10,5)
             DC      AL1(0,13,14,8,7,10,11,1,10,3,4,15,13,4,1,2)
             DC      AL1(5,11,8,6,12,7,6,12,9,0,3,5,2,14,15,9)
             DC      AL1(14,0,4,15,13,7,1,4,2,14,15,2,11,13,8,1)
             DC      AL1(3,10,10,6,6,12,12,11,5,9,9,5,0,3,7,8)
             DC      AL1(4,15,1,12,14,8,8,2,13,4,6,9,2,1,11,7)
             DC      AL1(15,5,12,11,9,3,7,14,3,10,10,0,5,6,0,13)
*

R0           EQU     0
R1           EQU     1
R2           EQU     2
R3           EQU     3
R4           EQU     4
R5           EQU     5
R6           EQU     6
R7           EQU     7
R8           EQU     8
R9           EQU     9
R10          EQU     10
R11          EQU     11
R12          EQU     12
R13          EQU     13
R14          EQU     14
R15          EQU     15
*
             END     CIPHER
*


